Example 8.15 Heat Conduction in a Rectangle with a Time Dependent Boundary Condition
Equation (8.1.35) is solved in Maple below for a general time dependent function, w(t), and plots are obtained for a particular step function.

```
> restart : with(inttrans) : with(plots) :
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t} u(x, t) = \frac{\partial^2}{\partial x^2} u(x, t) \quad \textbf{(1)}$$

```
> u(x,0):=0;
```

$$u(x, 0) := 0 \quad \textbf{(2)}$$

```
> bc1:=u(x,t)=0;
```

$$bc1 := u(x, t) = 0 \quad \textbf{(3)}$$

```
> bc2:=u(x,t)=w(t);
```

$$bc2 := u(x, t) = w(t) \quad \textbf{(4)}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) = \frac{d^2}{dx^2} U(x) \quad \textbf{(5)}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),laplace(w(t),t,s)=W(s),bc1);
```

$$bc1 := U(x) = 0 \quad \textbf{(6)}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),laplace(w(t),t,s)=W(s),bc2);
```

$$bc2 := U(x) = W(s) \quad \textbf{(7)}$$

```
> dsolve(eqs,U(x));
```

$$U(x) = \_C1\, e^{\sqrt{s}\, x} + \_C2\, e^{-\sqrt{s}\, x} \quad \textbf{(8)}$$

```
> U(x):=c[1]*cosh(s^(1/2)*x)+c[2]*sinh(s^(1/2)*x);
```

$$U(x) := c_1 \cosh(\sqrt{s}\, x) + c_2 \sinh(\sqrt{s}\, x) \quad \textbf{(9)}$$

```
> eq0:=eval(subs(x=0,bc1)):
> eq1:=eval(subs(x=1,bc2)):
> con:=solve({eq0,eq1},{c[1],c[2]}):
> U(x):=subs(con,U(x));
```

$$U(x) := \frac{W(s) \sinh(\sqrt{s}\, x)}{\sinh(\sqrt{s})} \quad \textbf{(10)}$$

```
> U(x):=factor(combine(simplify(U(x))));
```

$$U(x) := \frac{W(s) \sinh(\sqrt{s}\, x)}{\sinh(\sqrt{s})} \quad \textbf{(11)}$$

Next U1(x) is written as a product of two functions (see equations (8.1.31) and (8.1.33)):

```
> U1(x):=simplify(U(x)/W(s)/s);
```

$$U1(x) := \frac{\sinh(\sqrt{s}\ x)}{\sinh(\sqrt{s})\ s} \tag{12}$$

U1(x) is chosen so that q(s) is a higher order than p(s) (see equation (8.1.32)):

```
> G(s):=W(s)*s;
```

$$G(s) := W(s)\ s \tag{13}$$

where W(s) is the Laplace transform of the time dependent boundary condition w(t) in equation (8.11.35.). Next, U1(X) is inverted to the time domain as illustrated in section 8.1.7 to obtain f(t) as:

```
> P(s):=numer(U1(x));
```

$$P(s) := \sinh(\sqrt{s}\ x) \tag{14}$$

```
> Q(s):=denom(U1(x));
```

$$Q(s) := \sinh(\sqrt{s})\ s \tag{15}$$

```
> solve(Q(s),s);
```

$$0 \tag{16}$$

```
> _EnvAllSolutions := true;
```

$$\_EnvAllSolutions := true \tag{17}$$

```
> solve(Q(s),s);
```

$$-\pi^2\ \_Z1\sim^2,\ 0 \tag{18}$$

```
> 0,0,-n^2*Pi^2;
```

$$0,\ 0,\ -n^2\ \pi^2 \tag{19}$$

```
> mu0:=0;
```

$$\mu0 := 0 \tag{20}$$

```
> b[2]:=(s-mu0)^2*P(s)/Q(s);
```

$$b_2 := \frac{s \sinh(\sqrt{s}\ x)}{\sinh(\sqrt{s})} \tag{21}$$

```
> B[2]:=limit(b[2],s=0);
```

$$B_2 := 0 \tag{22}$$

```
> b[1]:=diff(b[2],s):
> B[1]:=limit(b[1],s=0);
```

$$B_1 := x \tag{23}$$

```
> A(s):=P(s)/diff(Q(s),s):
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := \frac{2 \sinh(\sqrt{\mu}\ x)}{\cosh(\sqrt{\mu})\ \sqrt{\mu} + 2 \sinh(\sqrt{\mu})} \tag{24}$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*n*Pi,mu^(3/2)=-I*n^3*Pi^3,mu=-n^2*Pi^2,A[n])):
> vars:={sin(n*Pi)=0};
```

$$vars := \{\sin(n\,\pi) = 0\} \quad \textbf{(25)}$$

```
> A[n]:=simplify(subs(vars,A[n])):
> A[n]:=simplify(subs(vars,expand(A[n])));
```

$$A_n := \frac{2\sin(n\,\pi\,x)}{\cos(n\,\pi)\;n\,\pi} \quad \textbf{(26)}$$

```
> b1s:=B[1]*subs(mu0=0,1/(s-mu0));
```

$$b1s := \frac{x}{s} \quad \textbf{(27)}$$

```
> b1t:=invlaplace(b1s,s,t);
```

$$b1t := x \quad \textbf{(28)}$$

```
> b2s:=B[2]*subs(mu0=0,1/(s-mu0)^2);
```

$$b2s := 0 \quad \textbf{(29)}$$

```
> b2t:=invlaplace(b2s,s,t);
```

$$b2t := 0 \quad \textbf{(30)}$$

```
> uns:=A[n]/(s-mu);
```

$$uns := \frac{2\sin(n\,\pi\,x)}{\cos(n\,\pi)\;n\,\pi\,(s-\mu)} \quad \textbf{(31)}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := \frac{2\sin(n\,\pi\,x)\;\mathrm{e}^{\mu\,t}}{\cos(n\,\pi)\;n\,\pi} \quad \textbf{(32)}$$

```
> unt:=subs(mu=-n^2*Pi^2,unt);
```

$$unt := \frac{2\sin(n\,\pi\,x)\;\mathrm{e}^{-n^2\,\pi^2\,t}}{\cos(n\,\pi)\;n\,\pi} \quad \textbf{(33)}$$

```
> f(t):=b1t+b2t+Sum(unt,n=1..infinity);
```

$$f(t) := x + \sum_{n=1}^{\infty} \frac{2\sin(n\,\pi\,x)\;\mathrm{e}^{-n^2\,\pi^2\,t}}{\cos(n\,\pi)\;n\,\pi} \quad \textbf{(34)}$$

Next, a step function is chosen for w(t) and plotted:

```
> w(t):=Heaviside(t-1)-1/2*Heaviside(t-2);
```

$$w(t) := \mathrm{Heaviside}(t-1) - \frac{1}{2}\,\mathrm{Heaviside}(t-2) \quad \textbf{(35)}$$

```
> plot(w(t),t=0..5,thickness=3,title="Figure Exp. 8.28.",axes=boxed);
```

Figure Exp. 8.28.



The Laplace transform of w(t) is:

```
> W(s):=laplace(w(t),t,s);
```

$$W(s) := \frac{1}{2}\,\frac{2\,e^{-s} - e^{-2\,s}}{s} \tag{36}$$

The function g(t) is obtained by inverting G(s):

```
> G(s):=s*W(s);
```

$$G(s) := e^{-s} - \frac{1}{2}\,e^{-2\,s} \tag{37}$$

```
> g(t):=invlaplace(G(s),s,t);
```

$$g(t) := \mathrm{Dirac}(t-1) - \frac{1}{2}\,\mathrm{Dirac}(t-2) \tag{38}$$

Next, the convolution integral is carried out to obtain the final time domain solution as:

```
> gtau:=subs(t=tau,g(t));
```

$$gtau := \mathrm{Dirac}(\tau - 1) - \frac{1}{2}\,\mathrm{Dirac}(\tau - 2) \tag{39}$$

```
> ftau:=subs(t=t-tau,f(t));
```

$$ftau := x + \sum_{n=1}^{\infty} \frac{2\sin(n\,\pi\,x)\,\mathrm{e}^{-n^2\pi^2(t-\tau)}}{\cos(n\,\pi)\,n\,\pi} \tag{40}$$

```
> U:=int(ftau*gtau,tau=0..t);
```

$$U := -\frac{1}{2}\,\mathrm{Heaviside}(t-2)\left(x + \sum_{n=1}^{\infty} \frac{2\sin(n\,\pi\,x)\,\mathrm{e}^{-n^2\pi^2(t-2)}}{\cos(n\,\pi)\,n\,\pi}\right) + \mathrm{Heaviside}(t-1)\left(x + \sum_{n=1}^{\infty} \frac{2\sin(n\,\pi\,x)\,\mathrm{e}^{-n^2\pi^2(t-1)}}{\cos(n\,\pi)\,n\,\pi}\right) \tag{41}$$

```
> u:=piecewise(t=0,0,t>0,subs(infinity=20,U)):
> plot3d(simplify(u),x=0..1,t=0..5,axes=boxed,title="Figure Exp. 8.29.",labels=[x,t,"u"],orientation=[135,45]);
```

Figure Exp. 8.29.



The dimensionless temperature at different points inside the rectangle are plotted as:

```
> plot([simplify(subs(x=1,u)),simplify(subs(x=0.75,u)),simplify(subs(x=0.5,u)),simplify(subs(x=0.25,u)),simplify(subs(x=0.0,u))],t=0..5.,thickness=3,axes=boxed,title="Figure Exp. 8.30.",labels=[t,"u"]);
```

Figure Exp. 8.30.



>